

9.8.6 Des program changes au clavier

Exceptionnellement, et contrairement aux applications précédentes, cet ensemble n'utilise pas de codes exclusifs, mais uniquement des messages MIDI note-on/note-off à des fins de télécommande. En guise d'introduc-

tion, voici quelques exemples d'applications de "remote control" MIDI, principe consistant à télécommander telle ou telle action à partir d'un clavier, en détournant de leur but initial les messages émis.

- **Télécommande d'un magnétophone**

En intercalant l'interface MTC-1 Fostex entre un clavier MIDI et le magnétophone huit pistes Fostex R8, les messages de notes enfoncées se substituent aux différentes fonctions du panneau avant du magnétophone : C2 pour play, C3 pour stop, C#3 pour record, etc.

- **Télécommande d'un séquenceur**

Le séquenceur Cubase procède de manière similaire, puisque n'importe quel clavier raccordé à l'entrée MIDI est capable de déclencher les fonctions start, stop, record, locate, cycle, etc. Pour ce faire, il convient tout d'abord de maintenir enfoncée une touche de contrôle (C6 par défaut), tout en appuyant sur celle qui correspond à la fonction désirée. Pour plus de souplesse, il est prévu de pouvoir reconfigurer la télécommande, c'est-à-dire d'en redéfinir les numéros de notes suivant ses désirs.

REMOTE CONTROL			
DISABLED			
REMOTE KEY	C6	QUANTIZE	D5
STOP		UNDO QUANTIZE	C#5
PLAY	>	NEXT SUBTRACK	C5
RECORD	>	DELETE SUBTRACK	A#4
FORWARD	>>	DELETE LAST V	B4
BEHIND	<<	QUANTIZE LAST V	A4
SOLO	D#5	NEXT TRACK	F5
CYCLE	E5	LAST TRACK	F#5
EXIT			

Figure 9.32 :
L'écran de
configuration de la
télécommande
Cubase.

- ***Télécommande d'un éditeur de son***

Certains des éditeurs bibliothécaires de la gamme Synthworks Steinberg offrent la possibilité de modifier les valeurs numériques éditables à l'écran, non plus seulement à partir de la souris, mais aussi par l'intermédiaire d'un ou plusieurs contrôleurs continus (molette de modulation, volume, etc.). Avec un contrôleur unique comme la modulation, il suffit de sélectionner le paramètre à éditer avant de l'altérer en déplaçant la molette, et ce en relatif (en plus ou en moins par rapport à sa valeur courante, en fonction du sens de déplacement du contrôleur), ou en absolu (suivant la position physique du contrôleur). Le Synthworks Proteus, pour ne citer que lui, est en mesure d'affecter seize contrôleurs continus à seize paramètres au choix. Lors de la programmation de cette table d'affectation, la fonction d'apprentissage (learn) reconnaît automatiquement le dernier contrôleur déplacé par l'utilisateur, lui évitant ainsi de devoir en saisir manuellement le numéro.

Ces trois exemples mettent en valeur l'aptitude d'un clavier à exploiter de simples messages dans le but de télécommander un appareil MIDI. Conçu dans le même esprit, le programme ci-après est destiné à faciliter l'écoute successive de l'ensemble des sons d'un instrument MIDI, en permettant de les sélectionner par l'intermédiaire d'un clavier de commande, l'appui sur certaines combinaisons de touches provoquant le calcul et l'envoi automatique de program changes.

Télécommande et program change

Le programme en question s'intercale entre la sortie d'un clavier MIDI et l'entrée d'un générateur de son (éventuellement celui de ce même clavier, à condition d'activer le mode local off). Outre le fait de renvoyer toutes les données reçues (fonction MIDI thru), ce programme, lorsqu'il détecte un appui sur les touches B5, A#5, A5 ou G#5 alors que C6 est déjà enfoncée, calcule et transmet un message de program change de la façon suivante :

C6 + B5 : numéro de programme = numéro de programme courant + 1
 C6 + A#5 : numéro de programme = numéro de programme courant - 1
 C6 + A5 : numéro de programme = numéro de programme courant + 10
 C6 + G#5 : numéro de programme = numéro de programme courant - 10

La molette de modulation (control change n° 1), dont la position est également convertie en un numéro de program change, sert à sélectionner grossièrement la zone à atteindre, pour l'affiner ensuite par l'intermédiaire des touches du clavier. Par souci de clarté, le numéro du programme courant est affiché à l'écran.

Les pièges à éviter

Tous les claviers MIDI ne traduisent malheureusement pas les actions d'enfoncement et de relâchement des touches par des messages identiques. Certains font appel au running status, d'autres transmettent une note-on de vélocité nulle en lieu et place d'une note-off. En se basant sur un clavier non dynamique réglé sur le canal 1, les exemples ci-après correspondent aux quatre combinaisons possibles traduisant les messages suivants :

- Enfoncement de la touche C6.
- Enfoncement de la touche B5.
- Relâchement de la touche B5.
- Relâchement de la touche C6.

1. note-off = 80H (sans running status) :

90H 60H 40H (note C6 enfoncée)
 90H 5FH 40H (note B5 enfoncée)
 80H 5FH 40H (note B5 relâchée)
 80H 60H 40H (note C6 relâchée)

2. note-off = note-on/vélocité 0 (sans running status) :

90H 60H 40H (note C6 enfoncée)
 90H 5FH 40H (note B5 enfoncée)

90H 5FH 00H (note B5 relâchée)
 90H 60H 00H (note C6 relâchée)

3. note-off = 80H (running status) :

90H 60H 40H (note C6 enfoncée)
 5FH 40H (note B5 enfoncée)
 80H 5FH 40H (note B5 relâchée)
 60H 40H (note C6 relâchée)

4. note-off = note-on/vélocité 0 (running status) :

90H 60H 40H (note C6 enfoncée)
 5FH 40H (note B5 enfoncée)
 5FH 00H (note B5 relâchée)
 60H 00H (note C6 relâchée)

Pour compliquer encore un peu plus l'organigramme de réception des données, ne perdons pas de vue le fait que d'autres notes sont susceptibles de s'intercaler entre l'appui et le relâchement des touches de télécommande.

Signification des variables

V50	numéro de program change		
V49	idem		
V99	dernier statut reçu	note-off	= 3
		note-on	= 1
		note-on/vélocité 0	= 2
		autre statut	= 0
V98	indicateur de note-on	C6 tenu	= 1
		C6 non tenu	= 0

Le programme (exécuté par activation du segment receive) du logiciel GenEdit

```

V50 = 0           : initialisation du program change
V99 = 0           : initialisation de l'indicateur de statut
V98 = 0           : initialisation de l'indicateur C6
GoSub L90         : saisie du canal pour program change
;
; Boucle principale
;
L00
GoSub L01         : réception d'un octet MIDI dans V00
V01 = V00 & %10000000 : est-ce une donnée ?
V01 = V01 EQ 0    : si oui
If V01 GoTo L02  : aller en L02
L21
V01 = V00 & %01100000 : si non, est-ce un statut de note?
V01 = V01 EQ 0    : si oui
If V01 GoTo L03  : aller en L03
V99 = 0           : si non, indicateur statut à zéro
V01 = V00 & %11110000 : est-ce un statut de control change ?
V01 = V01 NE SBO : si non
If V01 GoTo L00  : retour à la boucle
GoSub L01         : si oui, lire un octet
L20
V01 = V00 NE 1   : est-ce un octet de modulation ?
If V01 GoTo L00 : si non, retour à la boucle
GoSub L01         : si oui, lire la valeur de la modulation
TransmitVar V51   : transmettre le statut de program change
TransmitVar V00   : et l'octet de modulation comme numéro
TransmitVar V52   : transmettre la modulation à zéro
Transmit $01 $00 : pour qu'elle n'influe pas sur le son
TransmitVar V52   : transmettre
Transmit $7B $00 : un message "all notes off"
V50 = V00         : numéro de programme = valeur modulation
GoSub L75         : affichage écran du numéro de programme
V49 = V00         : V49 = numéro de programme
GoSub L01         : lire l'octet suivant
V01 = V00 & %10000000 : est-ce une donnée ? (running status)
V01 = V01 EQ 0    : si oui
If V01 GoTo L20  : reprendre en L20
GoTo L21         : si non, reprendre en L21
;
; Statut de note-on (90H) ou de note-off (80H)

```

```

:
L03
V01 = V00 & %00010000      : est-ce une note-on?
V01 = V01 EQ 16            : si oui
If V01 GoTo L04            : aller en L04
:
: note-off
:
V99 = 3                    : indicateur statut à 3
GoTo L00                   : retour à la boucle
:
: note-on
:
L04
V99 = 1                    : indicateur de statut à 1
GoTo L00                   : retour à la boucle
:
: Datas
:
L02
V01 = V98 EQ 1             : si indicateur C6 à 1 (note tenue)
If V01 GoTo L05            : aller en L05
V01 = V99 EQ 1            : si indicateur statut à 1 (note-on)
If V01 GoTo L06            : aller en L06
V01 = V99 EQ 3            : si indicateur statut à 3 (note-off)
If V01 GoTo L07            : aller en L07
GoTo L00                   : sinon (autre donnée), retour à la boucle
:
: Réception avec C6 encore tenu (V98 = 1)
:
L05
V01 = V99 EQ 3             : si indicateur statut à 3 (note-off)
If V01 GoTo L07            : aller en L07
V01 = V00                  : mémoriser l'octet V00 en V01
GoSub L01                  : lire l'octet suivant (vélocité) en V00
V02 = V01 EQ $5C           : si note = 5CH (G#5)
If V02 GoTo L08            : aller en L08
V02 = V01 EQ $5D           : si note = 5DH (A5)
If V02 GoTo L09            : aller en L09
V02 = V01 EQ $5E           : si note = 5EH (A#5)
If V02 GoTo L10            : aller en L10
V02 = V01 EQ $5F           : si note = 5FH (B5)
If V02 GoTo L11            : aller en L11
V02 = V01 NE $60           : si note différente de 60H (C6)

```

```

If V02 GoTo L14           : aller en L14
V99 = 2                   : si non, indicateur statut à 2 (note-on,
V98 = 0                   : vitesse 0), indicateur C6 à 0
GoTo L00                  : retour à la boucle
;
; Réception G#5/A5/A#5/B5 avec C6 encore tenu
;
L08
V50 = V50 - 10           : program change = program change - 10
GoTo L12                  : aller en L12
L09
V50 = V50 + 10           : program change = program change + 10
GoTo L12                  : aller en L12
L10
V50 = V50 - 1            : program change = program change - 1
GoTo L12                  : aller en L12
L11
V50 = V50 + 1            : program change = program change + 1
;
; Envoi du program change
;
L12
V02 = V00 NE 0           : si vitesse différente de 0
If V02 GoTo L15          : aller en L15
V50 = V49                 : sinon, restaurer le program change
V99 = 2                   : indicateur de statut à 2
GoTo L00                  : retour à la boucle
L15
V50 = V50 & %01111111    : forcer le program change sur 7 bits
TransmitVar V51           : en transmettre le statut
TransmitVar V50           : en transmettre la valeur
TransmitVar V52           : transmettre un statut de control change
Transmit $7B $00         : transmettre un message "all notes off"
V49 = V50                 : mémoriser le program change
GoSub L75                 : affichage écran du numéro de programme
GoTo L00                  : retour à la boucle
;
; Data note-on
;
L06
V01 = V00                 : mémoriser l'octet V00 en V01
GoSub L01                 : réception d'un octet (vitesse) en V00
V02 = V01 NE $60         : si note différente de 60H (C6)
If V02 GoTo L14          : aller en L14

```

```

V02 = V00 EQ 0           : si vitesse = 0
If V02 GoTo L13         : aller en L13
;
; Note on 60, vitesse <> 0 : V98 = 1
;
V98 = 1                 : indicateur C6 à 1 (note tenue)
GoTo L00                : retour à la boucle
;
; note-on 60, vitesse 0 : V98 = 0 / V99 = 2
;
L13
V98 = 0                 : indicateur C6 à 0 (note relâchée)
V99 = 2                 : indicateur statut à 2
GoTo L00                : retour à la boucle
;
; Note On <> 60
;
L14
V02 = V00 NE 0         : si vitesse non nulle
If V02 GoTo L00        : retour à la boucle
V99 = 2                 : si non, indicateur statut à 2
GoTo L00                : retour à la boucle
;
; Data note-off
;
L07
V01 = V00               : mémoriser l'octet V00 en V01
GoSub L01               : réception d'un octet (vitesse) en V00
V02 = V01 NE $60       : si note différente de 60H (C6)
If V02 GoTo L00        : retour à la boucle
V98 = 0                 : indicateur C6 à 0 (note relâchée)
GoTo L00                : retour à la boucle
;
; Réception/transmission d'un octet (mémorisation en V00)
;
L01
PTR = 0
ReceiveData 1 $F7 1     : réception d'un octet
V00 = PeekData 0 B      : mémorisation
TransmitVar V00         : transmission (fonction MIDI thru)
Return                  : retour
;
; Saisie canal pour program change (en V50)
;

```

